Experience With Wolfram Language in Teaching Electrodynamics

James Rohlf
Department of Physics
Boston University

Emphasis on readable code

Out[ ]=



$$\mathbf{E} = \frac{q}{4\pi\varepsilon_0 \mathcal{R}^2}\hat{\boldsymbol{\mathcal{R}}} = \frac{q}{4\pi\varepsilon_0 (\boldsymbol{\mathcal{R}}\cdot\boldsymbol{\mathcal{R}})^{3/2}}\boldsymbol{\mathcal{R}}$$

In[ ]:=

```
r = {x, y, z}; r′ = {x′, y′, z′}; ℛ = r - r′;
E = q / (4 π ε0 (ℛ.ℛ)^(3/2)) ℛ
```

There is some “work around”

1) The letter E is taken by Mathematica to be Euler’s constant.

In[ ]:=
```
E
```
Out[ ]=
```
ⅇ
```

In[ ]:=
```
E === ⅇ
```
Out[ ]=
```
True
```

Solution: use the Greek letter Ε, which looks nearly the same.

In[ ]:=
```
Ε === E
```
Out[ ]=
```
False
```

2) The prime symbol is taken by Mathematica to be differentiation.

In[ ]:=
```
f[x_] := x^2; f'[x]
```
Out[ ]=
```
2 x
```

Solution: use the Special Character symbol ′, which looks nearly the same.

In[ ]:=
```
x' === x′
```
Out[ ]=
```
False
```

The electric field for a moving charge is complicated! It depends on 3 vectors ($\mathcal{R}$, v, and a) and It has to be evaluated at an earlier time:

$$t - \frac{\mathcal{R}}{c}$$

$$E = \frac{q}{4\pi\,\varepsilon_0\left(\sqrt{\mathcal{R}.\mathcal{R}}\;c - \mathcal{R}.v\right)^3}\left(-(\mathcal{R}\,c - \mathcal{R}.v)\;\sqrt{\mathcal{R}.\mathcal{R}}\;a + \left(c^2 - v.v + \mathcal{R}.a\right)\left(c\,\mathcal{R} - \sqrt{\mathcal{R}.\mathcal{R}}\;v\right)\right)$$

Emphasis on units

```
In[∘]:= q = e;

r = RandomReal[1, 3] nm;

r′ = RandomReal[1, 3] nm;

ℛ = r - r′;

E = q / (4 π ε0 (ℛ.ℛ)^(3/2)) ℛ;

UnitConvert[E, V/m]

Out[∘]=
{-7.74942×10^8 V/m, -4.91396×10^9 V/m, -7.16616×10^8 V/m}
```

Fear no integral

Line of charge (textbook problem)

$$d\mathbf{E} = \frac{dq}{4\pi\varepsilon_0(\boldsymbol{\mathcal{R}} \cdot \boldsymbol{\mathcal{R}})^{3/2}}\boldsymbol{\mathcal{R}}$$

$$dq = \lambda dx'$$

$$\mathbf{E} = \frac{\lambda}{4\pi\varepsilon_0}\int_{-L/2}^{L/2} \frac{dx'\boldsymbol{\mathcal{R}}}{(\boldsymbol{\mathcal{R}} \cdot \boldsymbol{\mathcal{R}})^{3/2}}$$

In[∘]:=
```
Graphics[{{Opacity[0.5], Green, Line[{{.25, 0}, {0.27, 0}}]},
  Line[{{-0.01, -0.025}, {-0.01, 0.025}}],
  Line[{{1 + 0.01, -0.025}, {1 + 0.01, 0.025}}],
  Style[Text["r′", {0.375, -0.05}], 18, Bold ],
  Style[Text["r", {0.55, 0.25}], 18, Bold ],
  Style[Text["ℛ", {0.32, 0.25}], 18, Bold],
  Style[Text["𝒫", {0.5, 0.5}], 12], Style[Text["-L/2", {-.015, -0.09}], 12],
  Style[Text["L/2", {1.015, -0.09}], 12], Style[Text["ⅆq", {0.26, -0.05}], 12],
  {Thickness[0.005], Arrow[{{0.5, 0}, {0.5, 0.47}}],
   Arrow[{{0.26, 0}, {0.49, 0.47}}], Arrow[{{.5, 0}, {.25, 0}}]}, Thickness[0.02],
  {Opacity[0.3], Orange, Line[{{0, 0}, {1, 0}}]}, {Thickness[0.005], Blue,
   Arrow[{{0.5, 0}, {0.5, 0.1}}], Style[Text["ẑ", {0.55, -0.05}], 12],
   Style[Text["x̂", {0.47, 0.05}], 12], Arrow[{{0.5, 0}, {0.6, 0}}]},
  {Opacity[0.4], Darker[Green], Line[{{.25, 0}, {0.27, 0}}]}}]
```

Out[∘]=



In[∘]:=
```
r = {x, 0, 0}; r′ = {0, 0, z′}; ℛ = r - r′; $Assumptions = {L > 0, x ∈ ℝ, x ≠ 0};
```

$$\mathbf{E} = \frac{\lambda}{4\pi\varepsilon_0}\int_{-L/2}^{L/2}\frac{\mathcal{R}}{(\mathcal{R}.\mathcal{R})^{3/2}}\,dz' \;/.\; \lambda \to \frac{Q}{L}$$

Out[∘]=

$$\left\{\frac{Q}{2\pi x\sqrt{L^2+4x^2}\,\varepsilon_0},\ 0,\ 0\right\}$$

Off the symmetry axis, the problem is non-trivial.

Out[∘]=



In[∘]:= Clear[E]; r = {x, y, z}; r′ = {0, 0, z′}; ℛ = r − r′;

$$E[z'\_] = \frac{\lambda}{4\pi\varepsilon_0}\int \frac{\mathcal{R}}{(\mathcal{R}.\mathcal{R})^{3/2}}\, dz';\ (\text{*two steps for speed *})$$

$$E = \text{Simplify}\left[E\left[\frac{L}{2}\right] - E\left[-\frac{L}{2}\right] /.\ \lambda \to \frac{Q}{L}\right]$$

Out[∘]=

$$\left\{\frac{Q\,x\left(\frac{L-2z}{2\sqrt{L^2-4Lz+4(x^2+y^2+z^2)}}+\frac{L+2z}{2\sqrt{L^2+4Lz+4(x^2+y^2+z^2)}}\right)}{2L\pi(x^2+y^2)\varepsilon_0},\right.$$

$$\left.\frac{Q\,y\left(\frac{L-2z}{2\sqrt{L^2-4Lz+4(x^2+y^2+z^2)}}+\frac{L+2z}{2\sqrt{L^2+4Lz+4(x^2+y^2+z^2)}}\right)}{2L\pi(x^2+y^2)\varepsilon_0},\ \frac{Q\left(\frac{1}{\sqrt{L^2-4Lz+4(x^2+y^2+z^2)}}-\frac{1}{\sqrt{L^2+4Lz+4(x^2+y^2+z^2)}}\right)}{2L\pi\varepsilon_0}\right\}$$

In[∘]:= Series[E, {x, ∞, 2}] (*check result at large distance*)

Out[∘]=

$$\left\{\frac{Q}{4\pi\varepsilon_0 x^2}+O\left[\frac{1}{x}\right]^3,\ O\left[\frac{1}{x}\right]^3,\ O\left[\frac{1}{x}\right]^3\right\}$$

In[∘]:= Series[E, {y, ∞, 2}]
Series[E, {z, ∞, 2}]

Out[∘]=

$$\left\{O\left[\frac{1}{y}\right]^3,\ \frac{Q}{4\pi\varepsilon_0 y^2}+O\left[\frac{1}{y}\right]^3,\ O\left[\frac{1}{y}\right]^3\right\}$$

Out[∘]=

$$\left\{O\left[\frac{1}{z}\right]^3,\ O\left[\frac{1}{z}\right]^3,\ \frac{Q}{4\pi\varepsilon_0 z^2}+O\left[\frac{1}{z}\right]^3\right\}$$

In[∘]:= UnitConvert[E /. {Q → 1. nC, L → 1. cm, x → 1. cm, y → 2. cm, z → 1.25 cm, ε₀ → ε₀}, V/m]

(*numerical answer*)

Out[∘]=

{5362.27 V/m, 10 724.5 V/m, 6454.75 V/m}

In[ ]:=
```
Simplify[∇{x,y,z} × E](*check for zero Curl*)
```
Out[ ]=
```
{0, 0, 0}
```

In[ ]:=
```
$Assumptions = {L > 0, r > 0, z ∈ ℝ, 0 < ϕ < 2 π};
cyl = TransformedField["Cartesian" → "Cylindrical", √(E.E), {x, y, z} → {r, ϕ, z}] //
  Simplify(*check magnitude does not depend on ϕ*)
```
Out[ ]=

$$\frac{\sqrt{\frac{Q^2\left(L^4+L^2\left(8r^2-8z^2+\sqrt{L^4+8L^2(r^2-z^2)+16(r^2+z^2)^2}\right)-4(r^2+z^2)\left(-4r^2-4z^2+\sqrt{L^4+8L^2(r^2-z^2)+16(r^2+z^2)^2}\right)\right)}{L^2 r^2\left(L^2-4Lz+4(r^2+z^2)\right)\left(L^2+4Lz+4(r^2+z^2)\right)\varepsilon_0^2}}}{2\sqrt{2}\,\pi}$$

Sphere of charge

Out[ ]=



In[ ]:=
```
ClearAll["Global`*"];
$Assumptions = {r > 0, R > 0, z > 0};
r = {0, 0, z};
r′ = FromSphericalCoordinates[{R, θ′, ϕ′}];
ℛ = r - r′;
```

$$E[\theta'\_] = \frac{\sigma}{4\pi\varepsilon_0}\int\frac{R^2\,\mathrm{Sin}[\theta']\,\mathcal{R}}{(\mathcal{R}.\mathcal{R})^{3/2}}\,d\theta';$$

$$E[\phi'\_] = \int(E[\pi]-E[0])\,d\phi';$$

$$E = (E[2\pi]-E[0])\ /.\ \sigma\to\frac{q}{4\pi R^2}\ //\ \mathrm{Simplify}$$

Out[ ]=

$$\left\{0,\ 0,\ \begin{cases}\frac{q}{4\pi z^2\varepsilon_0} & R<z\\ 0 & \mathrm{True}\end{cases}\right\}$$

Ring of Current

Out[ ]=



In[ ]:=
```
ClearAll["Global`*"]; $Assumptions = {b > 0, ϕ′ > 0, r > b};
ℛ = r {Sin[θ], 0, Cos[θ]} - b {Cos[ϕ′], Sin[ϕ′], 0};
```

$$A = \frac{\mu_0\, \mathcal{I}}{4\,\pi}\ \text{AsymptoticIntegrate}\left[\frac{\text{b Cos}[\phi\prime]}{\sqrt{\mathcal{R}.\mathcal{R}}},\ \{\phi\prime,\ 0,\ 2\,\pi\},\ \{r,\ \infty,\ 2\}\right]\ \{0,\ 0,\ 1\}$$

Out[ ]=

$$\left\{0,\ 0,\ \frac{b^2\, \mathcal{I}\ \text{Sin}[\theta]\ \mu_0}{4\ r^2}\right\}$$

In[ ]:=

$$B = \text{Curl}\left[A\ /.\ \mathcal{I}\ b^2 \to m,\ \{r,\ \theta,\ \phi\},\ \text{"Spherical"}\right]$$

Out[ ]=

$$\left\{\frac{m\ \text{Cos}[\theta]\ \mu_0}{2\ r^3},\ \frac{m\ \text{Sin}[\theta]\ \mu_0}{4\ r^3},\ 0\right\}$$

In[ ]:=

$$\text{UnitConvert}\left[B\ /.\ \left\{m \to \left(10\ \text{cm}\right)^2\ \left(3\ \text{A}\right),\ r \to 1\ \text{m},\ \theta \to \frac{\pi}{6},\ \mu_0 \to \mu_0,\ B[[3]] \to 0\ \text{T}\right\},\ \text{T}\right]$$

Out[ ]=

$$\left\{1.63241943 \times 10^{-8}\ \text{T},\ 4.71238898 \times 10^{-9}\ \text{T},\ 0\ \text{T}\right\}$$

In[∘]:=
```
B = TransformedField["Spherical" → "Cartesian",
  B /. {m → 1, μ0 → 1}, {r, θ, ϕ} → {x, y, z}]
```

Out[∘]=

$$\left\{\frac{3\,x\,z}{4\left(x^2+y^2+z^2\right)^{5/2}},\ \frac{3\,y\,z}{4\left(x^2+y^2+z^2\right)^{5/2}},\ -\frac{x^2+y^2}{4\left(x^2+y^2+z^2\right)^{5/2}}+\frac{z^2}{2\left(x^2+y^2+z^2\right)^{5/2}}\right\}$$

In[∘]:=
```
StreamPlot3D[B, {x, -4, 4}, {y, -4, 4}, {z, -4, 4}]
```

Out[∘]=



Vector potential of current loop:

In[∘]:=

$$\texttt{Curl}\left[\frac{\mu_0\,\mathbf{Sin}[\theta]}{4\,\pi\,r^2}\,m\,\{0,0,1\},\ \{r,\theta,\phi\},\ \texttt{"Spherical"}\right]$$

Out[∘]=

$$\left\{\frac{m\,\mathrm{Cos}[\theta]\,\mu_0}{2\,\pi\,r^3},\ \frac{m\,\mathrm{Sin}[\theta]\,\mu_0}{4\,\pi\,r^3},\ 0\right\}$$

In[∘]:=
```
ClearAll["Global`*"]; m = π b^2 𝓘 {0, 0, 1};
r = {x, y, z};
```

$$\texttt{Simplify}\left[\texttt{Curl}\left[\frac{\mu_0\,m\times r}{4\,\pi\,(r.r)^{3/2}},\ \{x,y,z\}\right]\right] == \texttt{Simplify}\left[\frac{\mu_0}{4\,\pi}\left(\frac{3\,(r.m)\,r}{(r.r)^{5/2}}-\frac{m}{(r.r)^{3/2}}\right)\right]$$

Out[∘]=
```
True
```

In[∘]:=
```
ClearAll["Global`*"];
Simplify[TransformedField["Cartesian" → "Spherical",
  ∇{x,y,z} × (∇{x,y,z} × {0, 0, 1}/Sqrt[x^2 + y^2 + z^2]), {x, y, z} → {r, θ, ϕ}]]
```

Out[∘]=

$$\left\{\frac{2\,\mathrm{Cos}[\theta]}{\left(r^2\right)^{3/2}}, \frac{\mathrm{Sin}[\theta]}{\left(r^2\right)^{3/2}}, 0\right\}$$

In[∘]:=
```
Simplify[TransformedField["Cartesian" → "Spherical",
  -∇{x,y,z} (-(∇{x,y,z} · {0, 0, 1}/Sqrt[x^2 + y^2 + z^2])), {x, y, z} → {r, θ, ϕ}]]
```

Out[∘]=

$$\left\{\frac{2\,\mathrm{Cos}[\theta]}{\left(r^2\right)^{3/2}}, \frac{\mathrm{Sin}[\theta]}{\left(r^2\right)^{3/2}}, 0\right\}$$

In[∘]:=
```
A = {f[x, y, z], g[x, y, z], h[x, y, z]};
B = {p[x, y, z], q[x, y, z], r[x, y, z]};
∇{x,y,z} × (∇{x,y,z} × A) == ∇{x,y,z} (∇{x,y,z} · A) - ∇²{x,y,z} A
```

Out[∘]=

```
True
```

Visualization

In[∘]:=
```
ClearAll["Global`*"];
$Assumotion = L > 0;
r = {x, y, z};
r′ = {0, 0, z′};
ℛ = r - r′;
E[z′_] = λ/(4 π ε0) ∫ ℛ/(ℛ.ℛ)^(3/2) dz′; (*two steps for speed *)
E = Simplify[E[L/2] - E[-L/2] /. λ → Q/L];
```

In[ ]:=
```
SliceVectorPlot3D[E /. {Q → 1, ε0 → 1, L → 1},
 RegionBoundary[Cylinder[{{0, 0, -1}, {0, 0, 1}}, 1]],
 {x, y, z} ∈ Cylinder[{{0, 0, -1}, {0, 0, 1}}, 1],
 PlotStyle → Opacity[.2], BoxRatios → Automatic]
```

Out[ ]=



In[ ]:=
```
$Assumptions = {L > 0};
N[SurfaceIntegrate[E,
  {x, y, z} ∈ RegionBoundary[Cylinder[{{0, 0, -L}, {0, 0, L}}, L]]]]
```

Out[ ]=

$$\frac{1.\,Q}{\varepsilon_0}$$

Out[ ]=



In[ ]:=
```
ClearAll["Global`*"];
eqns = {∇²{x,y} u[x, y] == 0, u[0, y] == V, u[a, y] == V, u[x, 0] == 0, u[x, b] == 0};
sol = FullSimplify[DSolveValue[eqns, u[x, y], {x, y}]]
```

Out[ ]=

$$\sum_{K[1]=1}^{\infty} \frac{1}{\pi K[1]} 4 V \operatorname{Csch}\left[\frac{a \pi K[1]}{b}\right] \times \operatorname{Sign}[b] \times \operatorname{Sin}\left[\frac{\pi y K[1]}{b}\right] \operatorname{Sin}\left[\frac{\pi K[1]}{2 \operatorname{Sign}[b]}\right]^2 \left(\operatorname{Sinh}\left[\frac{\pi (a-x) K[1]}{b}\right] + \operatorname{Sinh}\left[\frac{\pi x K[1]}{b}\right]\right)$$

In[ ]:=
```
a = 2; b = 1; V = 1; Ω = Rectangle[{0, 0}, {a, b}];
solution = sol /. {∞ → 500} // Activate;
Plot3D[solution // Evaluate, {x, y} ∈ Ω,
 ImageSize → Medium, PlotRange → All, AxesLabel → {x, y, V}]
```

Out[ ]=



Electrodynamics 1

In[ ]:=
```
<< Notation`
Symbolize[ParsedBoxWrapper[SubscriptBox["_", "_"]]]
```

In[ ]:=
```
$Assumptions = {X ∈ Reals, Y ∈ Reals, z ∈ Reals};
c = 1; ω = 2 π; a = 2; b = 1;
```

$$k = \frac{1.}{c}\sqrt{\omega^2 - c^2\pi^2\left(\left(\frac{1.}{a}\right)^2 + \left(\frac{1.}{b}\right)^2\right)};$$

$$E_z = \text{Sin}\left[\frac{\pi x}{a}\right] \times \text{Sin}\left[\frac{\pi y}{b}\right];$$

$$E_x = \frac{i}{(\omega / c)^2 - k^2}\,(k\,\partial_x E_z);\ E_y = \frac{i}{(\omega / c)^2 - k^2}\,(k\,\partial_y E_z);$$

$$B_x = \frac{i}{(\omega / c)^2 - k^2}\left(-\frac{\omega}{c^2}\,\partial_y E_z\right);\ B_y = \frac{i}{(\omega / c)^2 - k^2}\left(\frac{\omega}{c^2}\,\partial_x E_z\right);$$

$$\mathbb{E} = e^{i\,(k\,z - \omega\,t)}\,\{E_x, E_y, E_z\}\ /.\ \{x \to X\,a,\ y \to Y\,b\};$$

$$B = c\,e^{i\,(k\,z - \omega\,t)}\,\{B_x, B_y, 0.\}\ /.\ \{x \to X\,a,\ y \to Y\,b\};$$

```
𝔼 = Re[ComplexExpand[𝔼, TargetFunctions → {Re, Im}]] // FullSimplify;
B = Re[ComplexExpand[B, TargetFunctions → {Re, Im}]] // FullSimplify;

Bs = Table[B, {t, 0., 1., 0.01}];
Es = Table[𝔼, {t, 0., 1., 0.01}];
```

In[∘]:=
```
plots = Table[Show[ListStreamPlot3D[
     Table[Es⟦i⟧, {X, 0., 1., 0.05}, {Y, 0., 1., 0.05}, {z, 0., 1.2, 0.06}],
     BoxRatios → {1, 0.5, 2.4}, StreamPoints → 100, StreamMarkers → "Arrow",
     StreamScale → {0.1, 20, 0.05}, PerformanceGoal → "Speed"],
    ListStreamPlot3D[Table[Bs⟦i⟧, {X, 0., 1., 0.05}, {Y, 0., 1., 0.05},
      {z, 0., 1.2, 0.06}], BoxRatios → {1, 0.5, 2.4},
     StreamPoints → 250, StreamColorFunction → Hue, StreamMarkers → "Line",
     StreamScale → {Full, 20}, PerformanceGoal → "Speed"]], {i, 1, 100}];
```

In[∘]:=
```
animation = ListAnimate[plots, AnimationRate → 5]
```

Out[∘]=



Electrodynamics 2

Out[ ]=



Pictures of Dynamic Electric Fields

ROGER Y. TSIEN
Department of Physics
Harvard University
Cambridge, Massachusetts 02138
(Received 24 June 1971)

In[∘]:=
```
ClearAll["Global`*"];
βmax = .95
sol = Solve[1/Sqrt[mf^2 + 1^2] == βmax, mf]; T = 1; t0 = 16 T;
mf = -sol[[All, 1, 2]].{0, 1};
x[t_] = Piecewise[{{-Sqrt[mf^2 + t^2] + Abs[mf], t < 0}, {0, t > 0}}];
β[t_] = Piecewise[{{-t/Sqrt[mf^2 + t^2], 0 < -t < T}, {T/Sqrt[mf^2 + T^2], -t > T}, {0, t > 0}}];
α[m_, t_] := 2 ArcTan[Sqrt[1 - β[t]]/Sqrt[1 + β[t]] Tan[(m - 1/2) π/36]];
nmax = 50;
ParametricPlot[Table[{x[t0 - n/nmax] + (n/nmax) Cos[α[m, t0 - n/nmax]],
   (n/nmax) Sin[α[m, t0 - n/nmax]]}, {m, 1, 36}], {n, 1, 2000}]
```

Out[∘]=

0.95

Out[∘]=







FIG. 12. Electric field lines of a charge decelerated from an initial velocity $\beta = 0.95$ to rest by a uniform force, $t_0 = 16\Delta t$.

In[ ]:=
```
ClearAll["Global`*"];
T = 1; t0 = 32 T; βmax = .2
sol = Solve[1/Sqrt[mf^2 + 1^2] == βmax, mf];
mf = -sol[[All, 1, 2]].{0, 1};
x[t_] := Piecewise[{{Sqrt[mf^2 + t^2] - Abs[mf], t > 0}, {0, t < 0}}];
β[t_] := Piecewise[{{t/Sqrt[mf^2 + t^2], 0 < t < T}, {T/Sqrt[mf^2 + T^2], t > T}, {0, t < 0}}];
α[m_, t_] := 2 ArcTan[Sqrt[1 - β[t]]/Sqrt[1 + β[t]] Tan[(m - 1/2) π/36]];
nmax = 25;
ParametricPlot[Table[{x[t0 - n/nmax] + (n/nmax) Cos[α[m, t0 - n/nmax]],
   (n/nmax) Sin[α[m, t0 - n/nmax]]}, {m, 1, 36}], {n, 1, 2000}]
```





FIG. 11. Electric field lines of a charge accelerated from rest by a uniform force to a final velocity $\beta = 0.20$, $t_0 = 32\Delta t$.

In[ ]:=
```
a = .01; β = .2; γ = 1/Sqrt[1 - β^2]; T = 2 π a / β;
α[R_, m_] := 2 ArcTan[Sqrt[1 - β]/Sqrt[1 + β] Tan[(m - 1 / 2) π/12 - γ β R/(2 a)]];
ParametricPlot[Table[{n Cos[(2 π n / T) + α[n , m]], n Sin[(2 π n / T) + α[n, m]]},
  {m, 12}], {n, 0, 4}, PlotRange → {-1, 1}]
```





FIG. 3. Electric field lines for a charge moving at tangential speed $\beta = 0.20$ on a circular path centered on the ×.

In[ ]:=
```
ω = 10; βmax = .1 ;
α[R_, m_] := 2 ArcTan[Sqrt[1 - βmax Cos[-ω R]]/Sqrt[1 + βmax Cos[-ω R]] Tan[(m - 1 / 2) π/24]];
ParametricPlot[Table[{n Cos[π/2 + α[n , m]], n Sin[π/2 + α[n, m]]}, {m, 24}],
 {n, 0, 4}, PlotRange → {-4, 4}]
```

Out[37]=





FIG. 13. Electric field lines of a charge undergoing one-dimensional simple harmonic motion with $\beta_{\max}=0.10$, $t_0=0$.

In[∘]:=

```
ω = 4; βmax = .5;

α[R_, m_, t_] := 2 ArcTan[Sqrt[1 - βmax Cos[-ω R + t]] / Sqrt[1 + βmax Cos[-ω R + t]] Tan[(m - 1/2) π / 24]];

Video["EFieldLines"] = ListAnimate@Table[
   ParametricPlot[Table[{n Cos[π/2 + α[n , m, t]], n Sin[π/2 + α[n, m, t]]}, {m, 24}],
    {n, 0, 4}, PlotRange → {-4, 4}], {t, 0, 2 π, .2}]
```

Out[∘]=